Recreational Puzzle Makers

Wilf LaLonde and John Pugh

Introduction

In this month’s column, we thought we would implement something relatively simple, a facility that can be used to construct crossword puzzles. Before we started, it wasn’t clear whether or not any inheritence would be needed. We also had no appreciation for the amount of computation needed to construct puzzles.

The Application (PuzzleMaker)

We began by implementing a relatively simple user interface using the window builder in Visual Smalltalk. This part of the application is Smalltalk dialect dependent, but we weren’t too concerned with the user interface design. We provided a facility (see Figure 1) that permits a user to type a word and then either click on “Add Across” or “Add Down”. The word is then added into the corresponding “across list” pull-down menu  (note word “bat”) or “down list” pull-down menu (note word “apple”). Each list can be viewed by clicking on the appropriate pull-down menus. We also permit the selected word in these lists to be deleted or migrated into the other list via the special arrow buttons that are between the two pull-down menus.



Class PuzzleMaker (see Listing 1) keeps track of two instance variables “changed” and “puzzleScaffold”. Each time we add, delete, or move a word, we set changed to true. When a user clicks on “Make”, changed is used to determine whether to perform a “reset” (which starts from scratch to find the first solution) or a set (which continues from the last solution). Variable puzzleScaffold contains an instance of a class with the same name (capitalized, of course). It maintains (1) a grid, a two-dimensional representation of the board which is currently set to extent 15 by 15 since we haven’t yet provide facilities in the user interface to change the size and (2) entries, the words converted to instances of a subclass of abstract class PuzzleEntry. We added facilities somewhat after the fact to save and retrieve puzzles from a file.



Normally, when we build a user interface driven application in a system that provides a window builder (as opposed to a visual programming environment as in VisualAge), we use the reaction/update design pattern which we discussed in a number of previous columns. In this case, however, we chose instead to use the simpler “use the widgets as local variables” design pattern, a pattern we typically use for dialogs but not more conventional applications. It’s a simpler approach that gets things going very fast when the user interface is not the main goal.



Normally, we avoid giving domain objects direct access to widgets because they can typically be passed as parameters (if needed). For example, to get the scaffold to print its grid on the puzzle widget (a graph pane), we ask it to “drawOn: aWidget”. This is in keeping with the basic philosophy of separating user interface concerns from domain object concerns. However, when domain objects take a long time to compute their results, there is often a need to provide status information about the state of the computation. If we don’t, a user is likely to make the assumption that the application is in an infinite loop. What we want to avoid is a group of interacting domain objects passing a status widget back and forth. So we ended up recording the status widget in the object itself. 

The PuzzleScaffold Class

The PuzzleScaffold class (see Listing 2) was designed to maintain two major pieces of information: (1) a two-dimensional grid which we implemented using a dictionary with points as keys, and (2) a collection of entries (instances of PuzzleEntry subclasses which are smart enough to be able to determine whether or not they could placed on top of the grid without causing a conflict; they can also remove themselves).



Secondary information includes (1) the grid extent (this would be part of the grid if we had access to a two-dimensional array facility), (2) the fonts to be used for displaying the word sequence numbers (smallFonts) and the words (largeFonts), (3) the status widget (statusWidget), (4) squareSize needed for displaying words (cached to avoid incessantly recomputing it), (5) pleaseStop which was intended to permit the computation to be halted (we didn’t finish this part), and finally (6) first and next which provides a linked list of successive coordinates (points) where the words can be placed on the grid.



Since we can’t lay out two successive words without an intervening space (horizontally and vertically), we actually create the entries so that they have an extra space at the ends. We additionally add a border around the grid containing one space. Consequently, although the visible part of each coordinate ranges from 1 to 15 (just to use specific numbers; any grid size is supported by the scaffold), the grid can be accessed with coordinates that range from 0 to 16.



When we start off, the grid is initialized to spaces (using method reinitializeGrid). Then each entry is successively asked to overlay itself on the grid (each entry tries all possibilities in the first/next linked list). If each entry succeeds, we have a solution. Otherwise, the last entry will have tried all possibilities (it is set to nil to indicate that there are no more possibilities). We back up one entry and ask it find the next possibility that works. If none work, we back up some more. Once an entry succeed, all subsequent entries start over from the beginning. 



In effect, given n entries and a grid with 225 elements (15 by 15), we are treating the collection of entries as an n-digit number each of which can be one of the 225 digits. We are using brute force to try out all possibilities.



In general, method reset is designed to start from the beginning and set is designed to continue from where it left off. After setting everything up, method reset actually uses set to do the real work. Method set looks a little messy because we added code to provide status information. We should probably extract the code dealing with displaying status information in a separate method but that would require additional instance variables.



Initially, we set up our linked list to provide coordinates row by row. That gave us poor looking solutions. We then tried a list of coordinates that started in the center and spiralled clockwise in an increasing radius around the center. Since we deliberately interleaved the “across” words with the “down” words (see method “acrossStrings: acrossStrings downStrings: downStrings”), this gave even worse solutions. For some reason, all “across” words ended up in the top-right half of the square and all “down” words in the bottom-left half. Finally, we tried randomizing the coordinates in the linked list. This seemed to work much better. Having tried three approaches, we kept them all and called it a strategy (so we added an instance variable to keep track of this strategy). 



To draw the grid (method “drawOn: aWidget”), we sort the “across” entries separately from the “down” entries. We want the highest entries before the lower ones and those more to the left before those to the right. This ordering is stored in the entries and is the order used for puzzle questions. Questions are currently not part of the application but should be added if databases of puzzles are to be ultimately kept. The numbers themselves, of course, are drawn in the puzzle. The entries then draw themselves using “drawUsingPen: aPen”; it turns out that extracting the pen from a widget is expensive so we only do it once.

The PuzzleEntry Classes

Class PuzzleEntry serves as an abstract class for classes PuzzleAcrossEntry and PuzzleDownEntry. The abstract class (see Listing 3) maintains five instance variables: scaffold, string, where, restoration, and order. 



Variable scaffold keeps track of the scaffold object which maintains the fonts to use, the extent, and the first/next linked list of coordinates, to name a few. Variable string (with blanks at both ends) is a puzzle word. Variable where indicates where the string is in the grid; it’s a two-dimensional coordinate (a point such as 2@3). Given a coordinate, the next possibility is obtained by method next (see listing) which simply obtains the subsequent entry in the first/next linked list maintained by the scaffold. Variable where is set to nil when there is no coordinate that it can be moved to. When a puzzle entry is told to reset or to set itself, the entry first obtains the characters in the grid that it is overwriting and stores it in restoration (provided that it is legal; it will not replace existing characters inserted by other entries). That way, it is easy to restore what was there when another choice is tried. Note that restoring a word like “apple” by replacing the corresponding grid entries by spaces is not adequate because some of the characters might have been added by other entries. Finally, variable order is used when drawing the word index using the small font; the word itself is drawn in the large font.



Concrete classes PuzzleAcrossEntry and PuzzleDownEntry (see Listings 4 and 5 respectively) provide a number of polymorphic messages including (1) stringCanFit (“across” words can only fall off the right end horizontally, “down” words can fall off the bottom vertically), (2) legitimate (“across” words cannot be in the top or bottom row, “down” words cannot be in the left or right column; remember the grid has an extra border around it), (3) “gridAt: index” and “gridAt: index put: anObject” (these methods allow the two dimensional grid to be accessed like a string; e.g., just as “self string at: 3” gives us the third character in the entry, “self gridAt: 3” gives us the third character in the grid where third means to the right for an “across” word and down for a “down” word), (4) firstCoordinate (this method provides the coordinates for the first non-blank character in the string; i.e., one to the right for “across” words and one down for “down” words), and (5) coordinatesAndCharacterDo: (this sequencing operation like keysAndValuesDo: for dictionaries provides successive non-blank coordinates and characters).



For viewing ease, each subclass determines the color to use for drawing (red for “across”and blue for “down”).

What We Learned

Actually, we knew that using brute force to try out all possibilities was going to be slow. And it is. Just for fun, we added words until we couldn’t get a solution after a minute. Then we let it run overnight. Ten hours later, it was still plugging away. Clearly, a better approximate algorithm is needed. Perhaps a little investigation will bring up a useful candidate. With or without a better heuristic, there are other things I would do in the next implementation. 



I would leave it to the implementation to determine which word to place “across” or “down” but I would have the user submit them in groups. I would bring up one user-interface for each group where the user could try out all possibilities (assuming each group is not too big). Then the main application would try to lay out each group as entities on their own. There is no need for multiple strategies.



I would provide the user with the capability to start (reset) or continue (set). Why second-guess the user? I would also abstract out the code that tries out all possibilities into a special object, perhaps called a Permutation object. This would make the code much more understandable. We would also be able to use it in other brute force algorithms that try out all possibilities. If I could, though, I would use a HeuristicPermutation class.



I am also wondering if there is a use for some kind of abstraction of the status widget. I suspect that the answer is no but perhaps I haven’t use enough of them yet.

Conclusions

We showed how we might implement a naïve puzzle maker. We are now in a better position for the next generation puzzle maker. Our guess is that we’ll need three generations before we have something truly practical. We probably have to get rid of brute force search entirely.

Figure 1   A running instance of PuzzleMaker.



Listing 1 The PuzzleMaker class.



class:								PuzzleMaker

superclass:						ViewManager 

instance variables:		changed puzzleScaffold 

pool dictionaries: 		ColorConstants 

										OperatingSystemConstants



class methods



window builder

wbCreated

    ^true



examples

example1

    "PuzzleMaker example1"

    PuzzleMaker new open



instance methods 



initializing

initialize

    super initialize.

    changed := true.

    puzzleScaffold := PuzzleScaffold new 

		extent: 15@15; 

		strategy: #random.



window builder

createMenus: aPane

createViews

  	… code not shown …



file event handlers

fileSave

    | fileName |

    fileName := (FileDialog new saveFile: '*.puz') file.

    fileName isNil ifTrue: [^self "user cancelled"].

    ObjectFiler

        dump: (Array with: self grid with: self fileSaveEntries)

        newFile: fileName

   

fileOpen

    | fileName gridAndEntries grid entries |

    fileName := (FileDialog new

        title: 'Open File';

        fileSpec: '*.puz';

        hideReadonly;

        open) file.

    fileName isNil ifTrue: [^self "user cancelled"].

    gridAndEntries := ObjectFiler 

		loadFromPathName: fileName.

    entries do: [:entry | entry scaffold: puzzleScaffold].

    puzzleScaffold grid: grid; entries: entries.

    self acrossListWidget 

		contents: self acrossStringsFromEntries.

    self downListWidget 

		contents: self downStringsFromEntries.

    self updateGraphics

 

file event handler support

grid

    ^puzzleScaffold grid



fileSaveEntries

    ^puzzleScaffold entries collect: [:entry |

        entry shallowCopy scaffold: nil; yourself]

 

acrossEntries

    ^puzzleScaffold entries select: [:entry |

        entry isKindOf: PuzzleAcrossEntry]

acrossStringsFromEntries

    ^self stringsFromEntries: self acrossEntries



downEntries

    ^puzzleScaffold entries select: [:entry |

        entry isKindOf: PuzzleDownEntry]

downStringsFromEntries

    ^self stringsFromEntries: self downEntries

   

stringsFromEntries: entries

    ^entries collect: [:entry |

        entry string copyFrom: 2 to: entry string size - 1]



activation event handling

activated

    self updateGraphics



clicked event handling

clickedStop

	“Not yet functional.”

    puzzleScaffold pleaseStop: true

  

clickedAddLeft

    self add: (self entryWidget contents) 

		toWidget: self acrossListWidget

clickedAddRight

    self add: (self entryWidget contents) 

		toWidget: self downListWidget

   

clickedDeleteLeft

    self deleteSelectionFromWidget: 

		self acrossListWidget

clickedDeleteRight

    self deleteSelectionFromWidget: 

		self downListWidget

   

clickedMoveLeft

    self moveSelectionFromWidget: self downListWidget 

		to: self acrossListWidget

clickedMoveRight

    self moveSelectionFromWidget: self acrossListWidget 

		to: self downListWidget



clickedMake

    CursorManager execute changeFor: [

        changed

            ifTrue: [

                puzzleScaffold

                    acrossStrings: self acrossListWidget contents

                        downStrings: self downListWidget contents;

                    reset.

                changed := false]

            ifFalse: [puzzleScaffold set]].

    self updateGraphics



event handling support

add: entry toWidget: widget

    | list |

    changed := true.

    entry isEmpty ifTrue: [^self].

    list := widget contents.

    list isNil ifTrue: [list := OrderedCollection new].

    list add: entry.

    list := list asSortedCollection asOrderedCollection.

    widget contents: list; selection: entry; text: entry

   

deleteSelectionFromWidget: widget

    | selection list |

    changed := true.

    selection := widget text.

    selection isEmpty ifTrue: [^self].

    list := widget contents.

    list remove: selection.

    widget contents: list; selection: nil; text: ''.



moveSelectionFromWidget: widget1 to: widget2

    | selection |

    changed := true.

    selection := widget1 text.

    selection isEmpty ifTrue: [^self].

    self deleteSelectionFromWidget: widget1.

    self add: selection toWidget: widget2



updating

updateGraphics

    puzzleScaffold statusWidget: self statusWidget.

    puzzleScaffold drawOn: self puzzleWidget



widget references

acrossListWidget

    ^(self paneNamed: 'acrossListWidget')

downListWidget

    ^(self paneNamed: 'downListWidget')

puzzleWidget

    ^(self paneNamed: 'puzzleWidget')

entryWidget

    ^(self paneNamed: 'entryWidget')

statusWidget

    ^(self paneNamed: 'statusWidget')



Listing 2 The PuzzleScaffold class.



class:								PuzzleScaffold

superclass:						Object

instance variables:		entries extent grid strategy first next 

										smallFont largeFont squareSize 

										pleaseStop statusWidget

pool dictionaries: 		ColorConstants 



class methods



comment

comment

    "The grid is constructed to have an extra border of

    spaces all around to take into account the fact that

    each puzzle entry is padded with a space at both

    ends."



instance creation

new

    ^super new initialize



initializing

initialize

    self initializeFonts.

    self pleaseStop: false

  

initializeFonts

    self smallFont: (Font

        face: 'Times New Roman'

        size: 8

        fixedWidth: false).

    self largeFont: (Font

        face: 'System 10 bold'

        size: 10

        fixedWidth: false)



reinitializeGrid

    "Create a grid with an extra border all around."

    grid := Dictionary new.

    0 to: self extent y + 1 do: [:y |

        0 to: self extent x + 1 do: [:x |

            grid at: x@y put: $ ]]



obvious get/set

extent

width

height

grid

grid: aDictionary

pleaseStop

pleaseStop: aBoolean

statusWidget

statusWidget: aWidget

entries

entries: anOrderedCollection

    … code not shown …



non-standard get/set

extent: aPoint

    extent := aPoint.

    self reinitializeGrid



entries

acrossStrings: acrossStrings downStrings: downStrings

    "Create the entries, alternate across and down."

    | acrossEntries downEntries |

    acrossEntries := acrossStrings collect: [:string |

        PuzzleAcrossEntry new

            string: ' ', string, ' ';

            scaffold: self].

    downEntries := downStrings collect: [:string |

            PuzzleDownEntry new

                string: ' ', string, ' ';

                scaffold: self].

    entries := OrderedCollection new.

    [(acrossEntries size + downEntries size) > 0] whileTrue: [

        acrossEntries notEmpty ifTrue: [

            entries add: acrossEntries removeFirst].

        downEntries notEmpty ifTrue: [

            entries add: downEntries removeFirst]].

   

squareSize

    ^squareSize

squareOffsetFor: aCoordinate

    ^(aCoordinate - 1) * squareSize



ordering

first

    ^first

nextAfter: previous

    ^next at: previous

  

link: previous to: after

    "Ignore attempts to link to itself, to out of bounds points,

    or to a prior element in the list."

    (previous = after or: [

    after notNil and: [

    (after x between: 0 and: self extent x + 1) not or: [

    (after y between: 0 and: self extent y + 1) not]]])

        ifTrue: [^previous].

    (next includesKey: after)

        ifTrue: [^previous].

    next at: previous put: after.

    ^after



ordering strategy

strategies

    ^#(centerOut topDown random)

strategy

    ^strategy

strategy: aSymbol

    "One of the symbols in strategies."

    strategy := aSymbol.

    self perform: self strategy



ordering support

centerOut

    … code not shown …

topDown

    … code not shown …

 

random

    | keys randomizedKeys index previous |

    "Make up the keys."

    keys := OrderedCollection new.

    0 to: self extent y + 1 do: [:y |

        0 to: self extent x + 1 do: [:x |

            keys add: x@y]].



    "Choose them in a random order."

    randomizedKeys := OrderedCollection new.

    [keys notEmpty] whileTrue: [

        index := Time randomFrom1To: keys size.

        randomizedKeys add: (keys at: index).

        keys removeIndex: index].



    "Link the randomized keys together."

    next := Dictionary new.

    previous := first := randomizedKeys removeFirst.

    [randomizedKeys notEmpty] whileTrue: [

            previous := self link: previous 

				to: randomizedKeys removeFirst].

    self link: previous to: nil



success/failure

succeeded

    ^entries last succeeded

failed

    ^entries last failed



controlling

reset

    "Compute first legitimate values."

    self reinitializeGrid.

    self resetEntriesFrom: 1.

    self lastSuccessfulEntry = entries size

        ifTrue: [^self].

    self set



set

    | start entry misfits trials oldMisfits choices |

    start := self lastSuccessfulEntry.



    choices := self width * self height.

    misfits := entries size - start.

    trials := 1.

    misfits timesRepeat: [trials := trials * choices].



    [true] whileTrue: [

        "Notifier yield."

        self pleaseStop ifTrue: [self pleaseStop: false. ^self].

        start = 0 ifTrue: [^self].

        entry := entries at: start.

        entry set.

        entry succeeded

            ifTrue: [self resetEntriesFrom: start + 1]

            ifFalse: [

                oldMisfits := misfits.

                misfits := (entries size - start + 1) max: misfits.

                misfits > oldMisfits ifTrue: [

                    trials := 1.

                    misfits timesRepeat: [trials := trials * choices]]].

        statusWidget contents: 'Words that don''t fit: ',

				misfits printString, ', choices to try: ', 

				trials printString.

        trials := trials - 1.

        start := self lastSuccessfulEntry.

        start = entries size ifTrue: [^self]]! !



controlling support

lastSuccessfulEntry

    entries size to: 1 by: -1 do: [:index |

        (entries at: index) succeeded

            ifTrue: [^index]].

    ^0

  

resetEntriesFrom: index1

    "Reset all entries. Fail all subsequent entries upon first failure. Return failure."

    | failed entry |

    failed := false.

    index1 to: entries size do: [:index |

        entry := entries at: index.

        failed

            ifTrue: [entry fail]

            ifFalse: [

                entry reset.

                failed := entry failed]].

    ^failed



fonts

smallFont

smallFont: aFont

largeFont

largeFont: aFont

    … code not shown …



smallFontOffset

    ^smallFont charSize - 1

largeFontOffset

    ^(self squareSize // 2) + (0@3).



drawing

drawOn: aWidget

    | pen |

    entries isNil ifTrue: [^self].

    squareSize := aWidget extent // self extent.

    self succeeded

        ifTrue: [

            self sortEntries.

            pen := aWidget pen.

            pen fill: ClrBlack.

            entries do: [:entry |

                entry drawUsingPen: pen]]

        ifFalse: [aWidget pen fill: ClrRed]



sortEntries

    self sort: (entries select: [:entry |

        entry isKindOf: PuzzleAcrossEntry]).

    self sort: (entries select: [:entry |

        entry isKindOf: PuzzleDownEntry])



sort: unsortedEntries

    | sortedEntries |

    sortedEntries := unsortedEntries

        asSortedCollection: [:entry1 :entry2 |

            entry1 where < entry2 where or: [

            entry1 where y < entry2 where y or: [

            entry1 where y = entry2 where y and: [

            entry1 where x < entry2 where x]]]].

    sortedEntries with: (1 to: sortedEntries size) 

		do: [:entry :index |

            entry order: index]



Listing 3 The PuzzleEntry class.



class:								PuzzleEntry

superclass:						Object 

instance variables:		where string restoration order scaffold

pool dictionaries: 		ColorConstants 



get/set

where

where: aPoint

string

string: aString

order

order: anInteger

scaffold

scaffold: aPuzzleScaffold

    … code not shown …



 

scaffold queries

next

    ^self scaffold nextAfter: self where



smallFont

    ^self scaffold smallFont

smallFontOffset

largeFont

largeFontOffset

squareSize

squareOffsetFor: aCoordinate

extent

width

height

	… re-routing like smallFont above …



gridding

grid

    ^self scaffold grid.



gridEntries

    "Returns the grid entries for this puzzle entry; either characters or nil."

    | entries |

    entries := Array new: self string size.

    1 to: entries size do: [:index |

        entries at: index  put: (self gridAt: index)].

    ^entries



gridEntries: entries

    "Sets the grid entries for this puzzle entry; either characters or nil."

    1 to: entries size do: [:index |

        self gridAt: index put: (entries at: index)].

 

stringCanOverlap: gridEntries

    gridEntries with: self string do: [:gridEntry :character |

		(gridEntry == character or: [gridEntry == $ ])

            ifFalse: [^false]].

    ^true



success/failure

fail

    self where: nil

  

succeeded

    ^self failed not

failed

    ^self where isNil



controlling

reset

    self where: self scaffold first.

    self add ifTrue: [^self].

    self continue



set

    self failed ifTrue: [^self].

    self remove.

    self continue



continue

    [true] whileTrue: [

        self where: self next.

        self failed ifTrue: [^self].

        self legitimate ifTrue: [

            self add ifTrue: [^self]]].



arranging

add

    "Returns if succeeded or not."

    (self stringCanFit) ifFalse: [^false].

    restoration := self gridEntries.

    (self stringCanOverlap: restoration) ifFalse: [^false].

    self gridEntries: self string.

    ^true

  

remove

    self gridEntries: restoration



drawing

drawUsingPen: pen

    | start |

    pen font: self largeFont.

    self coordinatesAndCharacterDo: [:aCoordinate :character |

            start := self squareOffsetFor: aCoordinate.

            "Make white background."

            pen

                fill: (start + 1 corner: start + self squareSize - 1)

                color: ClrWhite.

            "Draw large character."

            pen

                place: start + self largeFontOffset;

                centerText: (String with: character)].



    "Draw small entry order."

    pen font: self smallFont.

    start := self squareOffsetFor: self firstCoordinate.

    pen

        place: start + self smallFontOffset;

        displayText: self order printString.



Listing 4 The PuzzleAcrossEntry class.



class:								PuzzleAcrossEntry

superclass:						PuzzleEntry 

instance variables:		“none”

pool dictionaries: 		ColorConstants 



gridding

stringCanFit

    ^(self where x + self string size - 1) <= (self width + 1)

legitimate

    "Across entries can't be at y=0 or y=height+1."

    ^self where y between: 1 and: self height

 

gridAt: index

    ^self grid at: self where + ((index - 1)@0)

gridAt: index put: anObject

    self grid at: self where + ((index - 1)@0) put: anObject

  

printing

printOn: aStream

    aStream nextPutAll: self string; nextPutAll: 'across'.



drawing

drawUsingPen: pen

    pen foreColor: ClrRed.

    super drawUsingPen: pen

 

firstCoordinate

    "Provide the coordinates for the first non-blank characters."

    ^self where + (1@0)



coordinatesAndCharacterDo: aBlock

    "Provide coordinates and characters

    which are not blank."

    2 to: self string size - 1 do: [:index |

        aBlock

            value: self where + ((index - 1)@0)

            value: (self string at: index)]



Listing 5 The PuzzleDownEntry class.



class:								PuzzleDownEntry

superclass:						PuzzleEntry 

instance variables:		“none”

pool dictionaries: 		ColorConstants 



gridding

stringCanFit

    ^(self where y + self string size - 1) <= (self height + 1)

legitimate

    "Across entries can't be at x=0 or x=width+1."

    ^self where x between: 1 and: self width

   

gridAt: index

    ^self grid at: self where + (0@(index - 1))

gridAt: index put: anObject

    self grid at: self where + (0@(index - 1)) put: anObject

  

printing

printOn: aStream

    aStream nextPutAll: self string; nextPutAll: 'down'.



drawing

drawUsingPen: pen

    pen foreColor: ClrBlue.

    super drawUsingPen: pen



firstCoordinate

    "Provide the coordinates for the first non-blank characters."

    ^self where + (0@1)



coordinatesAndCharacterDo: aBlock

    "Provide coordinates and characters

    which are not blank."

    2 to: self string size - 1 do: [:index |

        aBlock

            value: self where + (0@(index - 1))

            value: (self string at: index)]





� PAGE �3�








